


```

ga rb3 , arn rb4 ; rb3 := Raddr; R:= 39 0 1 1;
b2: ck 0 ; R := R shift rb2;
b3: ab 0 ; R := R v tabel[rb3];
gr (rb3) ; tabel[rb3] := R;
hr s3 ; return
b1: qq 31.39
b4: qq 1.39
e
;
; procedure spejling
; qq addr m
;
; Fra Sørens kode
;
b b10
; 0000000000 0000000000 0000000012 3456789abc start
; 0000000000 0000000000 0000000021 34a9bc6578 facit
c2: grn rb1 , grn rb ; spejlingssekvens
arn (s1) , tk 22 ; 0000000000 0000000000 0000000012 3456789abc
; 0000001234 56789abc00 0000000000 0000000000
ga rb1 , tk 10 ; 56789abc00 0000000000 0000000000 0000000000
ck -6 , ga rb ; 0000005678 9abc000000 0000000000 0000000000
tk 10 , ck -2 ; 9abc000000 0000000000 0000000000 0000000000
; 009abc0000 0000000000 0000000000 0000000000
ab rb , ck -8 ; 009abc5678 0000000000 0000000000 0000000000
; 0000000000 9abc567800 0000000000 0000000000
ab rb1 , tk 5 ; 0000001234 9abc567800 0000000000 0000000000
; 012349abc5 6780000000 0000000000 0000000000
gr rb , mb rb2 ; 002000a000 6000000000 0000000000 0000000000
sc rb , tk 1 ; 02000a0006 0000000000 0000000000 0000000000
gr rb1 , arn rb ; 0103490bc5 0780000000 0000000000 0000000000
mb rb3 , sc rb ; 0100090005 0000000000 0000000000 0000000000
tk -1 , ab rb1 ; 0010009000 5000000000 0000000000 0000000000
; 02100a9006 5000000000 0000000000 0000000000
ab rb , tk -27 ; 02134a9bc6 5780000000 0000000000 0000000000
; 0000000000 0000000000 0000000021 34a9bc6578
gr (s1) , hr s2
b2: qq 136 t512 ; 0010001000 1000000000 0000000000 0000000000
b3: qq 273 ; 0100010001 0000000000 0000000000 0000000000
b: qq ; 0000005678
; 012349abc5 6780000000
; 0103490bc5 0780000000
; 0003400bc0 0780000000
;
b1: qq ; 0000001234
; 02000a0006
e
;
; procedure drejning
; qq addr m
;
b b10
c3: arn(s1)
ck -4 , gr rb1
tk -28 , ab rb1
mb rb2 , gr (s1)
hr s2
b1: qq
b2: udn (p) , udn (p)
e
;
; procedure indsæt
; qq tabeladdr s1
; qq addr m s2

```

```

;
b b10
c4:  arn(s2)    , gr  rb1    ; m1:=m
      it (s1)    , pa  rb7    ; rb7 := tabeladdr
      pa rb2     t 3        ; init i
                                ; for i := 1 step 1 until 3 do
b3:  arn rb1    , gr  rb4    ; begin m2:=m1;
      pa  rb5    t 2        ; init j
                                ; for j := 1 step 1 until 2 do
b6:  hs  rc5    ; begin indsæt1(tabel,m2);
b7:  qq
      qq  rb4
      hs  rc2    ; spejling
      qq  rb4
b5:  bt  2      t -1
      hv  rb6    ; end j;
      hs  rc3    ; drejning
      qq  rb1
b2:  bt  3      t -1    ; end i;
      hv  rb3
      hr  s3
b1:  qq          ; m1
b4:  qq          ; m2

```

```

e
;
;   procedure writeinteger({p},R);
;

```

```

b b20
c7:  gr  rb16   , pp  12
      hv  rb4    NZ
      sy  16    , hr  s1
b4:  pi  512t511 ; OA:=1
      arn rb19   , gr  rb18
b1:  pm  rb16   , dln rb18
      hv  rb2    NZA
      qq          IZA
b2:  gm  rb16   , ck  -10
      ga  rb3
      pa  rb3    t 16    LZ
b3:  sy  0      NOA
      pm  rb18   , dln rb15
      gr  rb18
      pp  p-1    , bs  p
      hv  rb1
      hr  s1
b15: qq  10.39
b16: qq  0      ; work
b18: qq
;   109876543210
b19: qq 100000000000.39

```

```

e
;
;   procedure find
;

```

```

b b20
c6:  pan rb1    t d1-1    ; for i:=0 step 1 until 127 do
b1:  gr  0      t 1      ; tabela[i]:=40 0;
      ac  (rb1)   IPA
      hv  rb1    NPA
      grn rb2
                                ; m:=0;
b5:  hs  rc1    ; mloop: if findes(tabela,m)
      qq  d1
      qq  rb2    ; then goto rb6
      hv  rb6    ; else

```

```

pi 0      t 1015      ; QA:=0;
arn rb2
hv rb8      NZ      ; if m=0 then
pi 8      t 1015    LKB ; QA:=KB;
hv rb9
b8: pan rb7  t d2-1      ; i:=-1;
b7: arn 0      t 1      IPA ; i:=i+1; R:=lovlige[i];
gr rb10     , mb rb2    ; v := R; R := v ^ m;
sr rb10     ; R := R-v;
hv rb11     NZ      ; if (v^m)=v then begin
arn rb10     ; R:= v;
ab 0 DX     ; R:= -, v;
mb rb2     , gr rb12   ; R:= m ^ (-, v);
hs rc1     ; if findes(tabela, m^-,v)
qq d1
qq rb12
hv rb9     ; then goto usikker;
b11: hv rb7     NPA ; end loop lovlige;
pi 8 t 1015 ; QA:=true;
b9: hv rb6     NQA ; usikker: if QA then
hs rc4     ; indsæt(tabela,m);
qq d1
qq rb2
hv rb6     LKA ; skip print if KA
sy 64     , arn rb2   ; writecr; R:=m;
hs rc7
b6: arn rb2     , ar rb3   ; m:=m+1;
gr rb2     , sr rb4   ; if m ≠ 4096 then
hv rb5     NZ      ; goto mloop;
hr s1     ; return
b2: qq     ; m
b3: qq 1.39
b4: qq 4096.39
b10: qq     ; v
b12: qq     ; m^-,v

```

e

```

; Layout for print sequence:
; qq <b>.3+<h>.7+<f1>.9+<d>.13+<n>.14+<bE>.17+<f2>.19+<g1>.23+<g2>.27+<g3>.31+<g4>.35
; 0 ≤ b ≤ 15   No. of significant digits
; 0 ≤ h ≤ 15   Digits before .
; 0 ≤ d ≤ 15   No. of decimals
; 0 ≤ f1 ≤ 3   sign
; 0 ≤ n ≤ 1    0 in front
; 0 ≤ bE ≤ 7   digits in exponent
; 0 ≤ f2 ≤ 3   sign on exponent
; 0 ≤ gx ≤ 15  grouping
b8: qq 4.3+ 4.7+0.13+1.14+ 4.23 ; {nddd}

```

[The following taken from print in HELP 3]

[NUMBER PRINT ROUTINE

page 1]

```

b a51      ;
          ;
a42: qq 0      , ck 0      ; entry address part 0≤x<1024
a50: qq 570    , t1 -30    ; entry address part -512≤x≤511
c9: pm 28     DV          ; entry integer
pm -11     D            ; entry fractional
c8: pa ra11 X      t485   ; entry real; numberpart:=true;
gs ra12 , gp ra13   ; save p; save s;
ga ra14 , gm ra23   ; store exp2; store numberpart;
pm (s) , arn s     ;
pm (s1)          NA ; get layout address
a14: psn 0      X      ; s:=exp2;

```

```

ps s11 , cl -20 ; s:=exp2+11;
tk 14 , ga ra17 ; unpack layout: b
tk 10 , pa ra16 ; exp10:=0;
a40: it 0 , pa ra4 ; expprinting:
pp 256 , ck -6 ; bE:=256;
ga ra18 , tk 10 ; h
ck -8 , ga ra19 ; f1
tk 10 , ck -6 ;
ga ra20 , tk 11 ; d
tk -20 , gt ra21 ; -n
tk 20 , ck -1 ;
ga ra22 , tk -6 ; bE+f2
ca 1 , pp 10 ; if bE=1 then minexp:=10;
ca 2 , pp 100 ; if bE=2 then minexp:=100;
tln 34 , ar ra43 ; group picture
gr ra8 , snn ra23 ; R:=-abs(numberpart);
pa ra4 V t-15 LZ ; if numberpart=0 then H:=-15
a49: nk ra46 XV ; reconversion: else x:=numberpart×21exp2;
gr ra26 , hv ra31 ; goto if numberpart=0 then L3
bs s-11 , hv ra47 ; else if s>0 then conversion1
; else conversion2;
a34: hv ra48 , sr ra26 ; round x2: R:=rounded x;
a17: pa 0 XVD t11 NO ; if -,overflow then b:=11
mt ra7 , hv ra49 ; else goto reconversion;
bs (ra4) , ntn (ra4) ; if H>0 then begin R:=0;
a21: qq (ra18) ; h:=h-H end else h:=h-n;
a19: pp 0 , gt ra13 ; p:=f1; a13incr:=R;
bs p509 , hv ra36 ; if p≠3 then goto count h
a11: arn 485 D t-485 ; exppart:=-,numberpart; numberpart:=false;
hs ra5 LT ; if exppart then write 10
bs (ra24), arn ra23 ; if b1>0^x<0 then
[21.9.1967 (10) PRINT, PAIR page 18]

```

[NUMBER PRINT ROUTINE

page 2]

```

arn -480 DV NT ; R:=-
arn 32 DV ; else R:=+
bs p510 , ck 10 ; if R<0^p<2 then R=small
hs ra37 , pp 3 ; write sign
a36: ;
a18: bt 0 t-1 ; count h:
hsn ra2 , hv ra36 ; write space before digits
bs p509 , hv ra11 ;
a13: pp 0 , ncn 0 ; restore p;
hsn ra1 , qq ;
bt (ra4) t-1 ; count H
hsn ra , hv r-1 ; write digits before point
arn ra38 , bs (ra20) ; if d>0 then
hs ra5 , it -1 ; write point;
a20: bt 0 , hh ra39 ; count d, write decimals
a12: ps 0 , xrn ; restore s; M:=0;
bs (ra24) , pm ra16 ; if b2>0 then M:=exp10;
can (ra22), hr s1 ; EXIT
gm ra23 , srn ra9 ;
pm (ra22) DX ;
ps 9 , hh ra40 ; goto expprinting;
a37: ca p , hh s ; if p=0^Raddr=0 then go back
gr ra26 , arn ra9 ; sign not counted in group
ac ra8 , arn ra26 ;
a5: bs (ra) , hv ra1 ; if b1>0 then write digit
mb ra44 , ga ra6 ;
a39: hvn ra10 , arn ra3 ; else if digit≠0 then write digit
a4: it 0 t1 ; count H else write 0
bs 0 , hvn ra5 ; if H<0 then write 0

```



```

b b10
;
;   procedure findlovlige;
;
b b20,e20
;
; findlovlig(m)
; qq addr m
; retur fundet
; retur ikke fundet
;
e1:  pa rb2      t d2-1      ; i:=-1;
b2:  arn 0       t 1         IPA ;loop: i:=i+1; R:=liste[i]; PA:=marks liste[i];
      sr (s1)                ; if R=m then
      hr s2                 LZ ;   return s+2
      hv rb2             NPA ; if -,PA then goto loop;
      hr s3                 ; return s+3
;
; tilføj lovlig(m)
; qq addr m
;
e2:  arn d2             IPA ; if liste[0]<0 then begin
      hv rb4             NT ;
      arn (s1)           ; R:=m;
      gr d2             MA ; liste[0]:=R Amarked;
      hv rb20           ; goto print end else
b4:  hs rel           ; begin findlovlig(m);
      qq (s1)
      hr s2                 ; findes allerede: return s2;
      acn (b2)          M ; fjern mærkebits fra sidste entry
      arn (s1)           ; R:=m;
      is (b2)
      gr s1             MA ; liste[i+1]:=R Amarked;
b20: hr s2             LKA ;print: skip print if KA
      sy 64                ; writecr;
      hs rc7             ; writeinteger(p,m);
      hr s2                 ; return
c10: arn rb3           ; procedure findlovlige; R:=-1;
      gr rd2             MA ; lovlige[0] := -1 Amarked;
      pa rb5      t d8-1   ;
b5:  arn d8-1 t 1      IPB ;for row:=1 step 1 until Bmarked do
      is (rb5) , is (s)
      it s-1 , pa rb6 ; b6 := addr first col in row row;
b6:  arn 0       t1      IQA ; for col1:=1 step 1 until Amarked do
      is (rb6) , it s-1
      pa rb7                 ; b7:=col1-1;
      grn rb8                ; m:=0;
b7:  arn 0       t 1      IRA ; for col2:=col1 step 1 until Amarked do
      ab rb8 , gr rb8 ; m:=m v tabel[row,col2];
      gr rb9                 ; m1:=m
      pa rb11 t 3
b15: pa rb10 t 2          ; for i:=1 step 1 until 3 do begin
      arn rb9 , gr rb13 ; for j:=1 step 1 until 2 do begin m2:=m1;
b14: hs re2
      qq rb13                ; tilføj lovlig(m2);
      hs rc2
      qq rb13                ; m2:=spejling(m2);
b10: bt 2      t -1
      hv rb14                 ; end j;
      hs rc3                 ; m1:=drejning(m1);
      qq rb9
b11: bt 3      t -1
      hv rb15                 ; end i
      hv rb7                 NRA ; end col2

```

```

        hv  rb6          NQA ; end col1
        hv  rb5          NPB ; end row;
        hr  s1           ; return

b3:  qq -1.39
b8:  qq                ; m
b9:  qq                ; m1
b13: qq                ; m2
e; findlovlige
;
;      main
;
b b20
c0:  zq                ; stop
     vy  32            ; select(32);
     zl                ; clock:=0.0;
     hs  rc10          ; findlovlige;
     zl      , grf rb2 ; b2:=clock;
     sy  49      , sy 64 ; writetext({<a}); writecr;
     vy  16      , sy 64 ; select(16); writecr;
     qq  rb1     , hs  rc8 ; write({nddddddddddd.dd},RF);
b3:  zq                ;loop: stop
     vy  32      , zl   ; select(32); clock:=0.0;
     hs  rc6
     zl      , grf rb2 ; b2:=clock;
     sy  49      , sy 64 ; writetext({<a}); writecr;
     vy  16      , sy 64 ; select(16); writecr;
     qq  rb1     , hs  rc8 ; write({nddddddddddd.dd},RF);
     hv  rb3          ; goto loop;
b1:  qq 14.3+12.7+2.13+1.14+12.23+ 2.27 ;{nddddddddddd.dd}
b2:  qq
e ;main
ec0

```